Maple-assisted proof of formula for A183777

Robert Israel

1 July 2018

There are $2^5 = 32$ configurations for a 1×5 sub-array. Consider the 32×32 transition matrix T with entries $T_{ij} = 1$ if the bottom row of a 2×5 sub-array could be in configuration i while the top row is in configuration j, and 0 otherwise. The following Maple code computes it.

```
> Configs:=  [seq(convert(2^5+i,base,2)[1..5],i=0..31)]:
> Compatible:= proc(i,j) local k;
   for k from 1 to 4 do
      if Configs[i][k]+Configs[i][k+1]+Configs[j][k]+Configs[j]
  [k+1] = 2 then return(0) fi
   od:
   1
  end proc:
> T:= Matrix(32,32,Compatible):
```

Thus for $n \geq 1$ $a(n) = v^T\, T^n v/2$ where v is a column vector with all entries 1.

```
> v:= Vector(32,1):
```

To check, here are the first few entries of our sequence.

```
> TV[0]:= v:
  for n from 1 to 50 do TV[n]:= T . TV[n-1] od:
> A:= [seq(v^%T . TV[n]/2,n=1..21)];
```

$$A := [85, 567, 3435, 21935, 136843, 864671, 5431499, 34228999, 215374371, 1356329167, 8537916907, 53757639287, 338436138739, 2130795330527, 13415038654331, 84459841870343, 531746080445251, 3347809160159215, 21077341291331531, 132700223506346903, 835462783565678867] \quad (1)$$

Now here is the minimal polynomial P of T, as computed by Maple.

```
> P:= unapply(LinearAlgebra:-MinimalPolynomial(T, t), t);
```

$$P := t \mapsto -16777216\,t - 5242880\,t^2 + 2856\,t^{23} - 2240\,t^{22} - 53964\,t^{21} + 28080\,t^{20} + 629760\,t^{19} - 240624\,t^{18} - 4820032\,t^{17} + 1520480\,t^{16} + 24946368\,t^{15} - 7093376\,t^{14} + t^{27} - 2\,t^{26} - 83\,t^{25} + 104\,t^{24} - 88418752\,t^{13} + 23384960\,t^{12} + 214699008\,t^{11} - 51674880\,t^{10} - 353518848\,t^9 + 72862720\,t^8 + 384975872\,t^7 - 61685760\,t^6 - 263405568\,t^5 + 28114944\,t^4 + 101711872\,t^3 \quad (2)$$

```
> degree(P(t));
```

$$27 \quad (3)$$

This turns out to have degree 27, but with the t^0 coefficient 0. Thus we will have

$0 = u\,P(T)\;T^n\,v = \sum_{i=1}^{27} p_i\, b(i+n)$ where p_i is the coefficient of t^i in $P(t)$. That corresponds to a homogeneous linear recurrence of order 16, which would hold true for any u and v, after a delay of 1. It seems that with our particular u and v we have a recurrence of order only 9, corresponding to a factor of P .

```
> empirical:= a(n) = 5*a(n-1) + 26*a(n-2) - 98*a(n-3) - 188*a(n-4)
```

```
+ 580*a(n-5) + 392*a(n-6) - 1024*a(n-7) - 256*a(n-8) + 512*a(n-9)
:
Q:= unapply(add(coeff((lhs-rhs)(empirical),a(n-i))*t^(9-i),i=0.
.9),t);
```

$$Q := t \mapsto t^9 - 5\,t^8 - 26\,t^7 + 98\,t^6 + 188\,t^5 - 580\,t^4 - 392\,t^3 + 1024\,t^2 + 256\,t - 512 \quad \textbf{(4)}$$

The complementary factor $R(t) = \frac{P(t)}{Q(t)}$ has degree 18, again with the lowest coefficient 0.

```
> R:= unapply(normal(P(t)/Q(t)),t);
```

$$R := t \mapsto \left(t^{17} + 3\,t^{16} - 42\,t^{15} - 126\,t^{14} + 652\,t^{13} + 1876\,t^{12} - 5256\,t^{11} - 13840\,t^{10} + 24608\,t^9 + 56496\,t^8 - 68320\,t^7 - 131872\,t^6 + 107776\,t^5 + 174016\,t^4 - 86656\,t^3 - 119808\,t^2 + 26624\,t + 32768\right)\,t \quad \textbf{(5)}$$

```
> degree(R(t));
```

$$18 \quad \textbf{(6)}$$

Now we want to show that $c(n) = u\,Q(T)\,T^n\,v = 0$ for all $n \geq 1$. This will certainly satisfy the recurrence

$$\sum_{i=1}^{18} r_i\,c(i+n) = \sum_{i=1}^{18} r_i u\,Q(T)\,T^{n+i} v = u\,Q(T)\,R(T)\,T^n\,v = u\,P(T)\,T^n\,v = 0$$

where r_i are the coefficients of $R(t)$. To show all $c(n) = 0$ it suffices to show $c(1) = ... = c(17) = 0$.

First we compute $w = v^T\,Q(T)$, then multiply it with the already-computed $T^n v$.

```
> UT[0]:= v^%T:
  for n from 1 to 17 do UT[n]:= UT[n-1].T od:
  w:= add(coeff(Q(t),t,j)*UT[j],j=0..degree(Q(t))):
> seq(w . TV[n],n=1..17);
```

$$0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0 \quad \textbf{(7)}$$

This completes the proof.